Maple Lecture 18. Symbolic and Automatic Differentiation

The distinction between a formula and a function is very important in this lecture. Formulas are differentiated symbolically, while automatic differentiation [1] produces derivative functions. Numerical differentiation is not considered here. This lecture corresponds to [2, Chapter 9].

18.1 Symbolic Differentiation

Symbolic differentiation is the calculation of the derivative of a formula, very much according to rules you learned in calculus. The Maple command for symbolic differentiation is diff, and its "inert" version is Diff.

```
[> expsin := exp(sin(x));
```

The "inert" version of diff, the Diff, simply echoes the command, while diff executes:

```
[> Diff(expsin,x) = diff(expsin,x);
```

We can compute the second, third, fourth, ... derivative applying diff repeatedly, but doing this 25 times is cumbersome. Therefore, Maple has the dollar operator:

```
[> x$10;                                  # shortcut to build a sequence
[> diff(expsin,x$10);                     # here we have the 10-th derivative
```

In many applications, what we wish to derive is not defined explicitly, but implicitly by an equation. To recall what implicit differentiation is, we first do it the long way on the equation of a circle $x^2 + y^2 = 1$. The equation "circle" relates y to x. We first tell Maple we view y as a function of x:

```
[> alias(y=y(x)):
[> circle := x^2 + y^2 = 1;
```

To get diff(y,x), we differentiate the defining equation. Remember: the differentiation of an equation is an equation! (Also do not forget that `diff` uses a remember table.)

```
[> equ := diff(circle,x);
```

Now we need to solve for the derivative of y with respect to x:

```
[> solve(equ,diff(y,x));
```

The short way goes like this :

```
[> implicitdiff(x^2+z^2=1,z,x);           # implicit differentiation
[> implicitdiff(x^2+z^2=1,z,x$3);         # 3-rd derivative
```

18.2 Automatic Differentiation

Automatic differentiation is the calculation of the derivative of a function; the result is again a function. The Maple command for automatic differentiation is D. We need automatic differentiation for two reasons. First: not every function can be represented by a nice formula. Second, even if there is a formula, we may have to deal with huge expression swell which renders the result of symbolic differentiation very difficult to use.

First we illustrate the difference between diff and D, taking the example `expsin` from above:

```
[> funexpsin := unapply(expsin,x);
[> derfunexpsin := D(funexpsin);
[> derfunexpsin(1.2);
```

For a function f in several variables, `D[2$3]`(f) gives the function which returns the third derivative with respect to the second variable.

It is instructive to look at the following commands:

```
[> diff(cos(t),t);                   # differentiation of the formula cos(t)
[> diff(cos,t);                      # wrong !
[> D(cos(t));                        # also wrong
[> D(cos);                           # differentiation of the cosine function
```

The following illustrates the avoidance of expression swell by automatic differentiation:

```
[> jf := z -> z^2 + 1/4;             # a quadratic map
[> f10 := jf@@10;                    # iterate the map ten times
```

The `f10` is a function which applies the map jf ten times. We can see the recipe for this function symbolically by evaluating at some symbol z:

```
[> sf10 := f10(z);                   # symbolic formula
```

Expanding `sf10` really leads to expression swell, so we do not do this here, but even differentiation only once, leads to a larger expression:

```
[> dsf10 := diff(sf10,z);            # symbolic differentiation
[> fsdf10 := unapply(dsf10,z);       # turn formula into function
[> fsdf10(0.3);                      # evaluate the derivative
```

If we are only interested in the value of the derivative, then we better apply automatic differentiation:

```
[> Df10 := D(f10);                   # automatic differentiation
[> Df10(0.3);                        # evaluate the derivative
```

Verify the results obtained by `fsdf10(0.3)` and `Df10(0.3)` and compare the sizes of the procedures:

```
[> eval(fsdf10); eval(Df10);         # compare sizes
```

18.3 Assignments

1. Give the Maple commands to get the value of the 7-th derivative of $e^{x^{10}+2}$ at $x = 1$.
   Also give the value you obtained.
2. Give the Maple command(s) to compute $\frac{\partial^8 f}{\partial^5 x \partial^3 y}$ for $f(x, y) = e^{2x+\cos(y)}$.
3. For $f := (x, y) \rightarrow \cos(xy)$, compute the derivative function which returns $\frac{\partial^5 f}{\partial^2 x \partial^3 y}$.
4. Consider the curve defined by $f(x, y) = 3 + 2x + y + 2x^2 + 2xy + 3y^2 = 0$.
   Locally on the curve we can view y as a function of x, i.e.: $y = y(x)$.
   Compute formulas for the first and second derivative of y with respect to x.
5. Compute the derivative of the function $f(x) = \min(x^2 + 1, 2x + 3)$.

References

[1] A. Griewank. *Evaluating Derivatives: Principles and Techniques of Algorithmic Differentiation*. SIAM, 2000.

[2] A. Heck. *Introduction to Maple*. Springer-Verlag, third edition, 2003.